




Science Translations Software Reference Reprints



More articles and software are available on the Internet at www.filetiger.com and www.graphcat.com.

Have Macro, Will Travel 10 Rules for Creating a Universal Macro

©Jerry Stern 1993, 2000, All Rights Reserved
As seen in WordPerfectionist, July 1993

Applies to: WordPerfect/DOS 5.1 (specifics), WP/DOS 6.x (useful), WP/Win(general concepts only)

When standardizing the software operations in an office, one of our greatest challenges is creating macros that will help speed the tasks unique to each of our companies. That challenge becomes far more difficult when your office is one of several in different cities or countries. Within a multinational company, even a company that tries to maintain internal standards for software, each office will have slightly different computing environments. The copies of WordPerfect are different on every computer, in three ways. Those variations, whether they are the language module in use, the setup options chosen, or the release date of the software, all affect how a macro displays and runs. Some of those changes are relatively unimportant, but many of them could crash your macro.

You cannot always take a macro from one computer to another, and still have it function as you expected it to. Programming in a macro language is not like programming in a standardized programming language like C or BASIC. It is possible to write a macro so that it will run on copies of WordPerfect with every combination of these factors. It is possible to program macros smart enough to test their software environments, and act accordingly. It is not always simple to create a universally-correct macro, but it is possible.

In this article, we'll show you what problems to watch out for when converting a macro to run anywhere, anytime. We'll also make up some rules about what to do in a macro, or more importantly, what **not** to do.

Release Dates

As WordPerfect 5.1 has evolved since November 6, 1989, two types of changes have been made in the program. One of these is bug fixes. The other changes have been minor improvements to the software. For example, the Border Options menu was added to the Format:Other menu in the 1/19/90 release. A macro that attempted to use that feature would crash when run on the original 11/6/89 release. Therefore, the original release of WordPerfect 5.1 is a standard environment in which all of your universal macros must run.

Rule 1.: Use ONLY the features of WP that were available in the 11/6/89 release.

You will need three tools to write and test a universal macro. First, you need an installed copy of the 11/6/89 release of WordPerfect 5.1. Second, you need the newest release, installed in a separate directory or on another computer. Third, you need a copy of WordPerfect's Enhancement disk. This is a listing of all the changes that have been made to WordPerfect 5.1 since the original release, including both program changes and changes to the manual. The list of program changes is known as ENHANCE.DOC. It lists, in alphabetical order by feature, the date and specifics of every change. It is this listing that will help you track down features that changed between your two standard versions of WP. The Enhancements disk is available directly from WP Sales for a shipping and disk fee.

In your Enhancements listing, look up Macros, and browse. You will find many new {SYSTEM} commands. For example, the {SYSTEM}entry command was added in the June, 90 release, and is one of the most powerful commands in the macro language. Using it, you can capture the text of any entry in a list of names, including a file name from the List Files screen, the printer name from the Select screen, or the form name from the Paper Size/Type list. There are other added commands as well, and using any of them will add major complications to your macro. Avoid them. We'll cover how to make the occasional exception to this rule under "How to Break the Rules," below.

International Editions

Each of the {SYSTEM} commands has a number as well as a name. The names are different in the different language modules. You cannot even depend on these names to be consistent between different English language modules, such as US and UK English. The numbers do not change. For example, {SYSTEM}29- will work in any language, but its equivalent English command, {SYSTEM}entry-, will not.

Rule 2.: Use the number equivalents to all {SYSTEM} commands.

I have heard many arguments between instructors of WordPerfect concerning which is better: teaching users the number keys on the menus, or teaching the mnemonics, or letters. Those who favor letters claim that the numbers will change at the next upgrade of WordPerfect, but that the letters will remain the same. Perhaps. But the letters change in every language module, while the numbers remain the same.

Rule 3.: Use numbers for ALL menu choices within a macro, NEVER letters.

Beyond Numbers

Some of the menus, such as [Print/Shift-F7] and [List Files/F5], contain more than nine entries, and the options beyond nine have substitute symbols. For example, in the Print menu, Select Printer has no number on screen, but it is the eighth entry, and can be chosen by pressing 8.

Rule 4.: For menu entries beyond the ninth, use the forward slash (/) for 10, backslash (\) for 11, the colon (:) for 12, and the asterisk (*) for 13. For 14 use a question mark (?), and for 15 use the plus (+) sign. (These substitute keystrokes were added in the 1/90 release.)

Measurements may also present a problem in International language modules. The abbreviations that we use for inches, centimeters, and points are not consistent between languages. WP recognizes other characters for measurement symbols.

Rule 5.: Use the forward slash (/) to indicate inches, the colon (:) for centimeters, the asterisk (*) for points, the question mark (?) for WordPerfect units, and the plus (+) for WP 4.2 units. (Added in the 6/90 interim release.)

Within the List Files screen, and any other screen containing a listing, you can start a Name Search by pressing N, even when it is not listed on screen as available. There is a special International equivalent for this function.

Rule 6.: To begin a Name Search in any screen containing a listing, use the equals (=) sign. (Added in the 1/90 release.)

Yes in Any Language

Sometimes, there are no obvious number substitutes. How does a macro clear the screen? The keystrokes are [Exit/F7], Yes or No to "Save document?" and No to "Exit WP?" In the 11/6/89 WP release, there was no other way to clear the screen--only the letters work. In the 1/19/90 release, WPCorp added the first International substitute. Instead of Y, 1 means Yes, and 2 selects No. However, in the 6/29/90 release, another change was made, and + works for Yes, and - for No. The numbers do not work in the new releases. (The change was made to prevent users from mistakenly pressing 1 instead of F1 during an Exit, and losing a document.)

The changes cause problems if your macro needs to answer a Yes/No question. Fortunately, there were no international editions of WordPerfect that do not accept either the 1/2 or +/- choices, so your macro programming logic in a yes/no choice can be stated like this:

Rule 7.: If your macro uses a feature that asks a Yes/No question, test the software to discover the interim release date. If 6/90 or later, use + or -; if 1/90 or 3/90, use 1 or 2; and if 11/89, use Y or N.

WordPerfect's macro language has no direct method for reading the interim release date. The EXIT macro in figure 1 shows one way to trick WordPerfect into giving up this information. First, we need to know if the release is 6/90 or newer, so we test a feature added in that release, the {SYSTEM}entry- or {SYSTEM}29- command. If this command is used in a current version of WordPerfect, while the cursor is on a list entry, such as the Paper Size/Type list, it will return the name of the highlighted entry. If that returned value is not an empty value, we know that the release date is 6/90 or newer, and we can use + and - for yes and no.

If the text string returned by the {SYSTEM}29- command is empty, and we know that the paper size list was not empty, then that event identifies the release date as earlier than 6/90. To discover whether the date is 11/89, we need to test a feature added in the 1/90 release. The EXIT macro attempts to enter the Border Options menu, which was added in 1/90, and then use the {SYSTEM}menu- command ({SYSTEM}13-) to identify the menu number that the macro is in. If the menu number is 403 (Border Options), then the release date of the software is 1/90 or 3/90, and yes and no are 1 and 2. If not, yes and no must be expressed as y and n. Notice that the first half of the macro, that assigns yes and no values to variables, can be placed in a setup area of a macro, while the variables can be used anywhere in the macro that a yes/no question must be answered.

Non-Programming Problems

When you format a page in a macro, do not assume that a printout will be on 8 1/2" by 11" paper. The standard letter-size sheet of paper outside North America is the A4 size, or 210mm by 297mm (slightly narrower and longer than the U.S. size). Make sure any planned printouts will fit on both sizes of paper.

Other formats will vary as well. Try not to interfere with the local standards for date formats and currency symbols. In addresses, allow for postal codes that include letters and are longer than U.S. ZIP codes, permit longer telephone numbers, and include extra address lines.

Setup Variations

Some of the settings in WordPerfect's [Setup/Shift-F1] menu can crash a macro. Look out for these, in particular.

Sort List Files [Setup/Shift-F1, 3, 4, 5]: If your macro uses the [List Files/F5] screen, test it with the unsorted directory option, added in the 3/92 release, turned both on and off. Also test with both short and long directory lists.

Menu Bars [Setup/Shift-F1, 2, 4, 8]: The {CHAR} and {PROMPT} statements are wonderful for creating dialog boxes and full-screen interfaces, when combined with the appropriate positioning commands. However, the positioning commands in WP are relative while in the document screen, and absolute while in other screens. This means that a message positioned on line 1, while the macro is in a document screen, will be moved down to the second line of the screen if the top-line menu bar is turned on, and to the third line if the separator bar is also turned on. However, while in the [List Files/F5] screen, the message will display on the top line of the screen no matter how the menu options have been set.

This downward shifting of your boxes and messages causes two problems. First, when the menu bar is turned on, if you create a display box while in the document screen, even with the display turned off, and then update it from [List Files/F5], the message will display either one or two lines too high. Second, when the menu bar and separator line are turned on, positioning the cursor in line 23 or 24 will display text off the bottom of the screen.

Rule 8.: When using positioning commands for screen displays, create and update your messages and boxes from the

same screen, even if the {DISPLAY OFF} command has been used.

Rule 9.: When using positioning commands for screen displays, don't use the bottom two lines of the screen.

Changing the Setup options from within your macro can become very tempting--it's easier than dealing with all the possible limitations and combinations. Don't do it. It's not an acceptable way of dealing with variety. If your macro cannot run with the current Setup options, find a way to detect whether the troublesome setting is turned on or off, and display an appropriate warning message or instructions on how to change the settings.

How to Break the Rules

Some of the new features that WordPerfect 5.1 acquired in interim releases are very powerful. Some are so powerful that the temptation to use them in macros will be irresistible. The {SYSTEM}entry- command that we described earlier is one of these new commands that makes entire new applications possible, but it was not added to WP until the 6/90 release. What to do? Cheat. But do it professionally.

As an example, let's say that the macro that we need to write will look in the Paper Size/Type listing, search for and select an envelope form, record the name of that form and display it in a box as a double-check, and then format the envelope. All of these steps are possible in any interim release of WP, except recording the form name.

When the macro searches the list and calls the {SYSTEM}29- command to record the list entry, the result may be a text variable that is zero characters long. That is our tip-off that the WP release date is before 6/90. Our macro code would look like Figure 2.

This technique will run on every version of WordPerfect, but on those releases dated before 6/90, it will only display a message, such as "Ask your computer tech. to upgrade your WordPerfect software." It does fulfill our last rule.

Rule 10.: If you must limit your macros to a particular release date or newer, or a particular language module, or a particular Setup, test for those conditions within the macro, and add appropriate error messages.

Conclusion

When writing a macro that will travel across borders, or even to another local office, don't assume that WordPerfect 5.1 is just one product. A good macro will always run on the copy of WordPerfect 5.1 on which it was written and tested. With some attention to technique, an excellent macro can run in any interim release, with any language module, and any setup options.

Jerry Stern is a WordPerfect Certified Resource, a Corel Solution Partner, a Lifetime Member of the Association of Shareware Professionals, and the author of Graphcat and FileTiger.

Figure 1: EXIT.WPM

```

{}EXIT.WPM-                {}Exits ANY release of ANY language of WP51-

{Format}27                {}Goto paper/size/type listing-

{ASSIGN}name-{SYSTEM}29-- {}Test system entry feature-
{Exit}{Exit}

{IF}{LEN}name->0-
  {ASSIGN}yes-+-          {}interim date 6/90 or newer-
  {ASSIGN}no---
{ELSE}
  {Format}48              {}Test for border options menu-
  {ASSIGN}name-{SYSTEM}13--
  {IF}"{VARIABLE}name-="403"-
    {ASSIGN}yes-1-        {}interim date 1/90 or 3/90-
    {ASSIGN}no-2-
    {Cancel}
  {ELSE}{ASSIGN}yes-y-    {}interim date 11/89-
  {ASSIGN}no-n-
  {END IF}{Cancel}{Cancel}
{END IF}

{Exit}{VARIABLE}no-{VARIABLE}yes-

```

Figure 2:

```

{}CHEAT.WPM-
{Format}27                {}enter listing-
=envelope{Enter}          {}Name search-
{ASSIGN}name-{SYSTEM}29-- {}record item name-
{IF}{LEN}name-=0-        {}test item name-
{GO}OldMessage-          {}If empty, go to message-
{ELSE}{Enter}{Exit}      {}otherwise, exit form menu-
{END IF}

{}Format envelope and quit-

{QUIT}
{LABEL}OldMessage-
{}Display Old version message and quit-

{QUIT}

```

©Jerry Stern 1994, 2000, All Rights Reserved